# $Id: PKGBUILD 85421 2013-03-01 10:41:47Z andrea $
# Maintainer: Timothy Redaelli <timothy.redaelli@gmail.com>
# Contributor: Rorschach <r0rschach@lavabit.com>

pkgname=vidalia
pkgver=0.2.21
pkgrel=2
pkgdesc="Controller GUI for Tor"
url="https://www.torproject.org/vidalia"
arch=('i686' 'x86_64')
license=('GPL')
depends=('geoip' 'qt4' 'tor')
makedepends=('cmake')
install=vidalia.install
source=(https://www.torproject.org/vidalia/dist/$pkgname-$pkgver.tar.gz{,.asc})
md5sums=('8d3079a66d79c79ccfefb929b820f9ac'
         'c735521ae9d231e974c753a5206c2c0f')

build() {
  cd "$srcdir"
  mkdir build
  cd build
  cmake -D CMAKE_INSTALL_PREFIX=/usr -DUSE_GEOIP=1 ../$pkgname-$pkgver
  make
}

package() {
  cd "$srcdir/build"
  make DESTDIR="$pkgdir/" install
  install -Dm644 ../$pkgname-$pkgver/doc/vidalia.1 "$pkgdir"/usr/share/man/man1/vidalia.1 || return 1
}
